

# gnuplot FAQ

この文書は、gnuplot バージョン 4 (最終リリースは 4.6.5) とバージョン 5 (2014 秋にリリース予定) の両方を対象としています。

この FAQ のバージョンは \$Revision: 1.53 \$, \$Date: 2014/08/21 15:05:33 \$ です。

## 目次

|          |                                        |          |
|----------|----------------------------------------|----------|
| <b>0</b> | <b>メタな質問</b>                           | <b>3</b> |
| 0.1      | このドキュメントはどこで手に入れますか                    | 3        |
| 0.2      | このドキュメントに関するコメントはどこに送れば良いですか           | 3        |
| <b>1</b> | <b>一般的な質問</b>                          | <b>3</b> |
| 1.1      | gnuplot とは何ですか                         | 3        |
| 1.2      | それはどこから来て、そしてなぜ gnuplot と呼ばれているのですか    | 3        |
| 1.3      | gnuplot は何を提供してくれますか                   | 4        |
| 1.4      | gnuplot はスクリプトに適していますか                 | 4        |
| 1.5      | gnuplot は私のコンピュータで動かすことができますか          | 4        |
| 1.6      | ライセンスについて                              | 4        |
| 1.7      | gnuplot は FSF や GNU プロジェクトと何か関係があるのですか | 5        |
| 1.8      | 更なる情報はどこで手に入りますか                       | 5        |
| <b>2</b> | <b>インストール</b>                          | <b>5</b> |
| 2.1      | gnuplot の最新バージョンは?                     | 5        |
| 2.2      | gnuplot はどこで手に入れますか                    | 5        |
| 2.3      | gnuplot の最新の開発版はどこで手に入れますか             | 5        |
| 2.4      | gnuplot は私のシステムではどうコンパイルすれば良いですか       | 6        |
| 2.5      | ドキュメントはどこですか、そしてそれをどう手に入れますか           | 6        |
| 2.6      | 動作するサンプル                               | 6        |
| 2.7      | どうしたら gnuplot を修正、またはパッチを当てられますか       | 6        |
| 2.8      | どのオプションが gnuplot にコンパイルされているかを知るには     | 7        |
| <b>3</b> | <b>動作 (Working with it)</b>            | <b>7</b> |
| 3.1      | ヘルプはどうやって手に入れますか                       | 7        |
| 3.2      | どうしたらグラフをプリントアウトできますか                  | 7        |
| 3.3      | どうしたらグラフをワープロソフトに取り込めますか               | 8        |
| 3.4      | どうしたら gnuplot のグラフを編集や後処理できますか         | 8        |
| 3.5      | どうしたら記号の大きさや線の太さなどを変更できますか             | 9        |
| 3.6      | グラフをアニメーション化できますか                      | 9        |
| 3.7      | どうしたら陰関数のグラフが書けますか                     | 9        |
| 3.8      | 2 つの曲線間の領域を塗りつぶすには                     | 10       |
| 3.9      | pm3d によるデータファイルの splot で何も表示されません      | 10       |

|          |                                                           |           |
|----------|-----------------------------------------------------------|-----------|
| 3.10     | 3次元データの2次元射影を描画するには                                       | 10        |
| 3.11     | pm3d 色地図/曲面上に dots/points の散布図を重ねるには                      | 10        |
| 3.12     | ラベル付きの等高線を描画するには                                          | 11        |
| 3.13     | pm3d 色地図/曲面グラフに等高線を重ねるには                                  | 11        |
| 3.14     | pm3d による色付きの多面体描画は                                        | 11        |
| 3.15     | 色地図をモノクロのプリンタでもちゃんと印字するようなパレットは                           | 12        |
| <b>4</b> | <b>欲しい機能 (Wanted features)</b>                            | <b>12</b> |
| 4.1      | gnuplot 4.6, 5.0 等で新しくなったことは                              | 12        |
| 4.2      | gnuplot は画像フォーマット用のドライバをサポートしていますか                        | 12        |
| 4.3      | gnuplot は隠線処理できますか                                        | 12        |
| 4.4      | gnuplot は棒グラフ/ヒストグラム/箱型グラフが書けますか                          | 12        |
| 4.5      | gnuplot は円グラフが書けますか、四半期グラフはどうですか                          | 13        |
| 4.6      | 一つのページに複数のグラフが書けますか                                       | 13        |
| 4.7      | gnuplot は一つのグラフで複数の y 軸を使えますか                             | 13        |
| 4.8      | 一つのファイルに gnuplot のコマンドとデータの両方を書けますか                       | 13        |
| 4.9      | ラベルにギリシャ文字や上/下付きの文字を使えますか                                 | 13        |
| 4.10     | どうしたらアクセント付きの文字を出力できますか                                   | 14        |
| 4.11     | 縦と横の軸を 1:1 にできますか                                         | 14        |
| 4.12     | 異なるサイズのテキストを一つのグラフで出力できますか                                | 14        |
| 4.13     | どうしたらデータ点をスキップできますか                                       | 14        |
| 4.14     | どうしたら n 点毎にデータを扱えますか                                      | 14        |
| 4.15     | どうしたら鉛直線を書けますか                                            | 14        |
| 4.16     | どうしたらデータファイルを描画できますか                                      | 15        |
| 4.17     | どうしたら multiplot の描画を replot できますか                         | 15        |
| <b>5</b> | <b>その他</b>                                                | <b>15</b> |
| 5.1      | バグを見つけたのですがどうしたら良いですか                                     | 15        |
| 5.2      | 私のプログラムで gnuplot のルーチンを利用できますか                            | 15        |
| 5.3      | どう追加や拡張がなされていますか、それはどう手に入られますか                            | 16        |
| 5.4      | 積分や FFT や IIR フィルタなどが欲しいのですが                              | 16        |
| 5.5      | "heavy-duty" データを gnuplot で使えますか、または gnuplot に変わるものはないですか | 16        |
| 5.6      | 私の対話型ターミナルでどのようにホットキーを使えばいいですか                            | 16        |
| 5.7      | gnuplot を他のシステムに移植、またはパッチを作りました。どうしたらいいですか                | 16        |
| 5.8      | gnuplot の次期バージョンの開発について知りたいのですがどうしたらいいですか                 | 17        |
| 5.9      | FAQ に含まれるような未解決の質問は?                                      | 17        |
| <b>6</b> | <b>うまく扱うには (Making life easier)</b>                       | <b>17</b> |
| 6.1      | どうしたら2つの関数を重ならないように書けますか                                  | 17        |
| 6.2      | どうしたら描画前にデータにフィルタをかけられますか                                 | 17        |
| 6.3      | どうしたら現在の設定を保存/再利用できますか                                    | 17        |
| 6.4      | どうしたら splot で曲線 (格子でない) を書けますか                            | 17        |
| 6.5      | どうしたら x-y 平面上で他の関数に制限されている $f(x,y)$ のグラフを書けますか            | 18        |
| 6.6      | どうしたらプログラムから gnuplot を呼び出せますか                             | 18        |
| 6.7      | Planck 定数 (h に横線) が必要なときはどうしたらいいですか                       | 18        |
| 6.8      | 太陽のような数学記号 (丸に点) が必要なときはどうしたらいいですか                        | 18        |
| 6.9      | 全く何も無いページを出力するには                                          | 18        |
| 6.10     | 境界の位置を丁度ある場所に合わせてグラフを作るには                                 | 19        |

|     |                                                                  |    |
|-----|------------------------------------------------------------------|----|
| 7   | 共通の問題                                                            | 19 |
| 7.1 | 助けて! 私のフォントが使えない。                                                | 19 |
| 7.2 | x11 または qt で <b>gnuplot</b> が描画ウィンドウを開きません。なぜですか                 | 20 |
| 7.3 | なぜ <b>gnuplot</b> はとても小さい数字のデータを無視するのですか                         | 20 |
| 7.4 | シェルやスクリプトから <b>gnuplot</b> を実行しても結果のグラフはスクリーン上で一瞬出るだけでそのあと出てきません | 20 |
| 7.5 | 書いた式 (1/3 など) が意味のない結果を出力してしまいます                                 | 20 |
| 7.6 | 私の出力が不完全です!                                                      | 21 |
| 7.7 | $\text{\LaTeX}$ -terminal の出力は $\text{\LaTeX}$ の実行でエラーが出ます       | 21 |
| 7.8 | パイプで <b>gnuplot</b> を呼び出したり <b>gnuplot</b> スクリプトを使ってグラフが書けません    | 21 |
| 8   | 帰属                                                               | 21 |

## 0 メタな質問

### 0.1 このドキュメントはどこで手に入れますか

このドキュメントの最新版は [www.gnuplot.info/faq/](http://www.gnuplot.info/faq/) の Web 上にあります。

### 0.2 このドキュメントに関するコメントはどこに送れば良いですか

コメント、提案等は電子メールで開発者向けメーリングリスト [gnuplot-beta@lists.sourceforge.net](mailto:gnuplot-beta@lists.sourceforge.net) に送ってください。faq.tex に関する提案をお寄せください。このファイルは [gnuplot.cvs.sourceforge.net/viewvc/gnuplot/faq/](http://gnuplot.cvs.sourceforge.net/viewvc/gnuplot/faq/) にあります。

## 1 一般的な質問

### 1.1 gnuplot とは何ですか

**gnuplot** はコマンド駆動型の対話型関数描画プログラムです。関数の描画、およびデータ点の描画を 2 次元、3 次元の両方で、様々な異なる書式で行うことができます。これは主に科学的なデータを画面表示するものとして設計されています。**gnuplot** の著作権は保護されていますが、自由に配布でき、代金を払う必要はありません。

### 1.2 それはどこから来て、そしてなぜ **gnuplot** と呼ばれているのですか

**gnuplot** の作者は Thomas Williams, Colin Kelley, Russell Lang, Dave Kotz, John Campbell, Gershon Elber, Alexander Woo、そして多くの協力者です。

以下は Thomas Williams によります:

私は微分方程式の講義を、そして Colin は電磁気学の講義を取っていて、2 人ともそれらに関する数学が視覚化できればなあと考えていました。私達はある EE (Engineering Electronics) VLSI 研のシステム管理者として働いていて、そして、グラフィック端末とコーディングを行なう時間がありました。その投稿は我々の期待以上に良く受け入れられ、そして、不完全でしたが、ファイルデータに関するいくつかのサポートに我々を駆り立てました。

GNUplot という紹介はどんなものも正しくはありません。このプログラムの本当の名前は "gnuplot" です。ずいぶん "Gnuplot" と書かれているのを見ますが、それは我々の大半が、固有名詞やタイトルであっても、文章を小文字で開始することに嫌悪を持っている

からです。gnuplot は GNU プロジェクトや FSF とは、極わずかの意味を除けば少しも関係ありません。我々のソフトウェアは完全に独立にデザインされたもので、そして "gnuplot" という名前は実は妥協によるものです。私は "llamaplot" と呼びたかったし、Colin は "nplot" と呼びたかったのです。そして、我々は "newplot" がいいだろうということで合意しましたが、しかしそのときその名前を持つ、絶対的におぞましい Pascal のプログラムを計算機科学部でたまに使っていることを知りました。それで私は "gnuplot" がいいごろ合わせになるとして、後で Colin もそれに同意したのです。

### 1.3 gnuplot は何を提供してくれますか

- 2次元での関数やデータの、たくさんのスタイル(点、折れ線、誤差表示)による描画
- 3次元でのデータや曲面のたくさんのスタイル(等高線描画、網かけ)による描画
- 整数、実数、複素数での代数演算
- Marquardt-Levenberg 最適法を用いたデータ駆動型の曲線回帰
- 多くのオペレーティングシステム、多くのグラフィックファイル形式出力、そして多くの出力装置のサポート
- 広範囲にわたるオンラインヘルプ
- ラベル、タイトル、軸、データ点への  $\text{\TeX}$  ライクな書式による見出し付け
- 対話型入力形式のエディット機能とヒストリ(履歴)機能のサポート(多くのプラットフォーム上で)

### 1.4 gnuplot はスクリプトに適していますか

ええ。gnuplot は対話型実行中に追加コマンドを含むファイルを読み込むことができますし、既に存在するファイルや標準入力からのコマンド列をパイプを使ってバッチモードでそれを処理することもできます。gnuplot は、Octave のような高水準な数学パッケージの背後で実行するグラフィックドライバとして使われていますし、cgi スクリプトでラップすることで容易に Web 起動型の描画生成ツールとして使うこともできます。

### 1.5 gnuplot は私のコンピュータで動かすことができますか

gnuplot は、MS Windows, linux, unix, OSX などの多くのプラットフォーム上で、広く利用できます。現在のソースコードは、VMS, Ultrix, OS/2, MS-DOS, BeOS, Macintosh などの古いシステムのサポートも保持していますが、バージョン 4.0 以降では古いプラットフォームでは徹底的なテストは行われていません。

gnuplot のソースは、それなりに標準的 (ANSI/ISO C, POSIX 等) な環境から多少外れたものの上でもコンパイルできるでしょう。

### 1.6 ライセンスについて

gnuplot はボランティアの集団によって作られたものです。彼らは gnuplot、およびその使用の許可、不許可に関する法的な文書を作成できません。また何の保証もありません。自分自身の責任で使用してください。以下は、R. Freund による数学関数のサブルーチンパッケージの README からの引用です:

全ての意図、目的に関して、そのコードがやることに対する任意の記述はそのコードが我々のマシンの上で昨年のある火曜日にたまたま行なったことに対して我々が考えたこと、と解釈されるべきです。もしついていけばそのコードは貴方に対してもある日同じことをするでしょう。繰り返しますが「それ」をついているだけだと本当に思えますか？

## 1.7 gnuplot は FSF や GNU プロジェクトと何か関係があるのですか

gnuplot は FSF が作ったのでも FSF がメンテナンスしているのでもありません。一時期 FSF から配布されていたこともあります、しかし現在はこれは正しくありません。gnuplot は、GNU General Public License (GPL) に全く保護されていません。

gnuplot は、無償であると言う意味でフリーウェアです。しかし、あなたが改変したものを自由に配布することが認められるか、ということに関してはフリーウェアではありません。Copyright というファイルの modification (改変) と redistribution (再配布) の項目を読んでそれを承認してください。

## 1.8 更なる情報はどこで手に入りますか

gnuplot Web ページ [www.gnuplot.info](http://www.gnuplot.info) を見てください。

英語以外の他の言語での文書や入門書もあります。最新のリストについては、[gnuplot.sourceforge.net/help.html](http://gnuplot.sourceforge.net/help.html) の "Localized learning pages about gnuplot" の節をご覧ください。

# 2 インストール

## 2.1 gnuplot の最新バージョンは？

gnuplot の最新の正式リリース版は、2012 年 3 月リリースの 4.6 です。更新バージョン (パッチレベル 1, 2, ...) は、だいたい 6 ヶ月毎にリリースされます。gnuplot の開発版は、現在は 5.0 となっています。

## 2.2 gnuplot はどこで手に入れられますか

最初に最も適切な場所は [www.gnuplot.info](http://www.gnuplot.info) です。ここから、例えば SourceForge 上の開発プロジェクトサイト [sourceforge.net/projects/gnuplot](http://sourceforge.net/projects/gnuplot) などの、色々なサイトの場所を見つけられるでしょう。

ソース配布物 ("gnuplot-4.6.0.tar.gz" または同様の名前) は公式の配布サイト [sourceforge.net/projects/gnuplot](http://sourceforge.net/projects/gnuplot) にあります。

## 2.3 gnuplot の最新の開発版はどこで手に入れられますか

gnuplot の開発版は、cvs ソースツリーの形で、オンラインで直接 [sourceforge.net/projects/gnuplot/](http://sourceforge.net/projects/gnuplot/) の "CVS" セクションから参照できます。そこに含まれているドキュメントに従って、例えば、以下のようなコマンド列によって最新のソースをすべてダウンロードすることもできます。

```
cvs -d:pserver:anonymous@gnuplot.cvs.sourceforge.net:/cvsroot/gnuplot login
cvs -z3 -d:pserver:anonymous@gnuplot.cvs.sourceforge.net:/cvsroot/gnuplot co -P gnuplot
```

または (bash では)

```
export CVSROOT=:pserver:anonymous@gnuplot.cvs.sourceforge.net:/cvsroot/gnuplot
cvs login
cvs -z3 checkout gnuplot
```

パスワードを聞かれたら `<return>` を打ちます。

その後、gnuplot のコンパイルの段階の `./configure` の前に、`./prepare` を実行して最新の configure 用のファイルを作成する必要があります。

公式の gnuplot の開発版の暫定的なバイナリリリースはありませんので、自分自身でコンパイルする必要があります。しかし、OS/2, Windows, Macintosh などのいくつかのプラットフォーム用の非公式なバイナリ版はどこかに見つかるかも知れません。

重要な注意: 開発版に対する質問は必ず [gnuplot-beta@lists.sourceforge.net](mailto:gnuplot-beta@lists.sourceforge.net) に送ってください。

## 2.4 gnuplot は私のシステムではどうコンパイルすれば良いですか

他のソフトのインストールと同様です。README.1ST, README, INSTALL ファイルを読んでください。

- Unix では、`./configure` (単一ユーザのインストールならば `./configure --prefix=$HOME/usr`) を使い、`make` を実行し、最後に `make install` または `make install-strip` とします。後者はデバッグ情報を除いた、より小さな実行ファイルをインストールします。RPM パッケージを作りたいなら、最後のステップを `checkinstall` かまたは `checkinstall make install-strip` で置き換えますが、あなたのマシンで `checkinstall` パッケージをサポートしている必要があります。
- Windows では、makefile は `config/mingw`, `config/msvc`, `config/watcom`, `config/cygwin` 内にあります。makefile の先頭に書かれているオプションを修正し、その makefile が置かれているのと同じディレクトリ内で適当な `make` ツールを実行してください。追加説明も、makefile に書かれています。
- DOS では、bash か DJGPP を使っているなら `djconfig.sh` を実行するだけです。
- 他のプラットフォームでは、適切な makefile (例えば OS/2 なら `makefile.os2`) を `config/` ディレクトリから `src/` にコピーして、makefile のヘッダにあるオプションを必要なら更新して、`src` にディレクトリを移動して、`make` を実行してください。

## 2.5 ドキュメントはどこですか、そしてそれをどう手に入れられますか

ドキュメントはすべてソース配布物に含まれています。個々のセクションは、gnuplot の作業中に `help` キーワード を打つことで見ることができます。サブディレクトリ `docs` と `tutorial` の中を見てください。そこには、PDF や HTML 版のユーザマニュアルを生成するファイルがあります。

[gnuplot.sourceforge.net/documentation.html](http://gnuplot.sourceforge.net/documentation.html) に、そのコピーが置いてあります。

## 2.6 動作するサンプル

ソース配布物には動作するサンプルのディレクトリが含まれています。これらサンプル、およびその結果の描画グラフは [gnuplot.sourceforge.net/demo/](http://gnuplot.sourceforge.net/demo/) でも見ることができます。

## 2.7 どうしたら gnuplot を修正、またはパッチを当てられますか

このためには `gnuplot` をコンパイルし直す必要があります。

様々な人々が作った修正は、出力ドライバのようなファイルを置き換えることで行なうか、またはパッチ当てで行ないます。ファイルの置き換えの場合、それ用の README ファイル、またはそのファイルの最初の行に何か情報があるかも知れません。

パッチを当てる場合は、`patch` コーティリティが必要で、もしかしたら `automake`, `autoconf` というツールも必要かもしれません。パッチを適用する場合の典型的なやり方は `patch -p0 <newfunctionality.diff` です。

寄稿されたパッチの置き場所は gnuplot の sourceforge サイト [sourceforge.net/tracker/?group\\_id=2055&atid=302055](http://sourceforge.net/tracker/?group_id=2055&atid=302055) の "Patches" セクションにあります。

## 2.8 どのオプションが gnuplot にコンパイルされているかを知るには

コンパイル済みの gnuplot が与えられている場合、show コマンドで、それがコンパイルされたときに使用されたコンパイルオプション (別名コンパイレイションオプション、ビルドオプション) の一覧を表示できます。

```
gnuplot> show version long
```

## 3 動作 (Working with it)

### 3.1 ヘルプはどうやって手に入れますか

このドキュメントを読んでください。

初期プロンプトで help コマンドを実行してください。その後キーワードをたどって行ってください。plot や set から始めるのがいいでしょう。

もし、持っているならマニュアルを読んでください

demo サブディレクトリを見てください。ヒントを与えてくれるでしょう。

システム管理者、または gnuplot を設定した同僚に尋ねてください。

それら全てに失敗したら、gnuplot の最新版に更新する、あるいは更新するようシステム管理者に圧力をかけてください。そして質問をニュースグループ [comp.graphics.apps.gnuplot](http://comp.graphics.apps.gnuplot) に投書するか、またはメーリングリスト [gnuplot-info@lists.sourceforge.net](mailto:gnuplot-info@lists.sourceforge.net) にメールを送ってください。そのメーリングリストが別に受けてしまう圧倒的な量 spam を避けるため、投稿の前にそのメーリングリストに参加 (subscribe) しなければいけません。参加の方法は、gnuplot のマニュアル本文にあります。メールを送る際は、バージョン番号とオペレーティングシステムを併記することを忘れないでください。そのメーリングリストを購読したいならば、URL [lists.sourceforge.net/lists/listinfo/gnuplot-info](http://lists.sourceforge.net/lists/listinfo/gnuplot-info) を参照してください。しかしニュースグループ [comp.graphics.apps.gnuplot](http://comp.graphics.apps.gnuplot) が直接読めるならば、そうしないでください。質問をそのニュースグループに投稿すれば、それはメールでの返事を請求する、あるいは要約を投稿する良い形式と見なされます。

### 3.2 どうやったらグラフをプリントアウトできますか

生成される出力の種類は set terminal コマンドで定義します。例えば

set terminal postscript はグラフを PostScript 形式で出力します。出力は set output コマンドでリダイレクトできます。

以下の例は、まず sin(x) のグラフをスクリーン上に描画し、同じグラフを PostScript 出力ファイルとして再描画します。

```
gnuplot> plot [-6:6] sin(x)
gnuplot> set terminal postscript
Terminal type set to 'postscript'
Options are 'landscape monochrome "Courier" 14'
gnuplot> set output "sin.ps"
gnuplot> replot
gnuplot> unset output          # 出力先をデフォルトに戻す
gnuplot> unset terminal        # 出力形式をデフォルトに戻す
gnuplot> system("print sin.ps") # PS ファイルをプリントアウト (環境依存)
gnuplot>
```

プラットフォームに依存しない出力形式の復帰は、コマンド `set term push/pop` によって以下のように行なえます。

```
gnuplot> set terminal postscript eps color lw 15 "Helvetica" 20
gnuplot> set out 'a.eps'
gnuplot> replot
gnuplot> set term pop
```

`set term pop` コマンドは、その前に対応する `set term push` を実行していない場合は、起動時の出力形式復帰します。`unset terminal` を実行した場合も同じです。

対話型出力形式の中にはそのツールバー上にプリンタアイコンを提供するものもあります (`win`, `wxt`, `qt` 等)。それは、他の出力形式を使用するのではなく、システムに用意されている一般的なシステムツールを使い、現在のグラフを印刷するか、またはファイルに保存します。すなわち、印刷メニューで "save to png" を選択して得られるファイルは、`set term png; replot;` によって得られるファイルとは違うものになります。

### 3.3 どうやったらグラフをワープロソフトに取り込めますか

基本的に、グラフをそのワープロソフトが理解できる形式のファイルに保存すれば (`set term` と `set output` を使います。上を見てください) そのワープロソフトで読み込むことができるでしょう。後で正しいサイズに変換もできるという点でベクトル形式のフォーマット (PostScript, emf, svg, pdf,  $\text{T}_\text{E}\text{X}$ ,  $\text{L}\text{A}\text{T}_\text{E}\text{X}$  など) の方がいいでしょう。

細かいことはそのワープロソフトに依存します。`set term` として使用できるファイル形式のリストを見てください。

多くのワープロソフトは Encapsulated PostScript (\*.eps) のグラフを取り込めます。**gnuplot** では、`set terminal postscript eps` を使うことで eps 出力を選択できます。注意しておきますが、EPS ファイルのグラフの bounding box を調べ、修正するといいいでしょう (手動、あるいは `gnuplot` Web ページにある `fixbb` スクリプトによって)。いくつかの (大抵の?) ワープロソフトは、EPS ファイルに含まれる実際の画像を表示しませんので、表示イメージをあなた自身で追加する必要があります。その目的のためには、GSView ビューワ (OS/2, Windows, X11 用があります)、あるいはいくつかの Unix 上の ps ツールが使えます。

OpenOffice.org 等の Windows のオフィスアプリケーションには、EMF 形式のベクトル画像を処理できるものがあります。そのような画像ファイルは emf 出力形式でも作れますし、windows 出力形式のグラフウィンドウのツールバーから 'Save as EMF...' を選択することでも作れます。

OpenOffice.org は SVG 形式も、AutoCAD の dxf 形式も読みこむことができます。

`gnuplot` を使って  $\text{T}_\text{E}\text{X}$  や  $\text{L}\text{A}\text{T}_\text{E}\text{X}$  文書に取り込むためのグラフを作る方法は何通りもあります。取り込む \*.tex ファイルの断片を直接生成する出力形式もありますし、`\includegraphics` コマンドを使って取り込むための \*.eps, \*.pdf, \*.png 出力を生成するものもあります。epslatex と cairolatex 出力形式は、画像ファイル (\*.eps か \*.pdf) とそれを参照する \*.tex 文書ファイルの両方を生成します。`gnuplot` 4.6 には tikz 出力形式があり、出力を pdflatex で処理する場合はこれでテキストとグラフ全部を生成できます。

大抵のワープロソフトはビットマップ画像 (png, pbm 等) を取り込むことができますが、この方法の短所は、そのグラフが `gnuplot` によって生成されたときのサイズによってその解像度が制限されてしまうことです。一般に、文書が最終的にプリントアウトされる解像度よりもその解像度はかなり低くなります。

mif 出力形式は FrameMaker 用の出力を生成します。

### 3.4 どうやったら `gnuplot` のグラフを編集や後処理できますか

これはあなたが使う出力形式に依存します。

- **svg** 出力形式 (scalable vector graphics) の出力は **Inkscape** ([www.inkscape.org](http://www.inkscape.org)), **Skencil** ([www.skencil.org](http://www.skencil.org)), **Dia** ([projects.gnome.org/dia/](http://projects.gnome.org/dia/)), のような **svg** エディタで更なる編集も行なえますし、**OpenOffice.org** (OO.o Draw プリミティブへの変換によって) の中に取り込むこともできます。
- PostScript や PDF 出力は、Adobe Illustrator や Acrobat のようなツールで直接編集できますし、**pstoedit** パッケージで他の色々な編集可能なベクトル形式に変換することも可能です。pstoedit は [www.pstoedit.net](http://www.pstoedit.net) にあります。
- mif 出力形式は FrameMaker で編集できる文書を生成します。
- DXF 出力形式は AutoCAD の書式で、他のいくつかのアプリケーションでも編集できます。
- ビットマップ形式の画像 (png, jpeg, pbm 等) は ImageMagick や Gimp の様なツールで編集することができます。一般に、ベクトルグラフィック形式の後処理を行なうにはベクトルグラフィックプログラム (ドロー系ツール) を、ピクセル画像にはピクセル画像系のプログラム (ペイント系ツール) を使う必要があります。

### 3.5 どうやったら記号の大きさや線の太さなどを変更できますか

gnuplot は、例えば色や太さや点の形などの、線や点の色々な属性を設定するための様々なコマンドを提供しています。コマンド `test` は、現在選択されている出力形式用の、利用可能な定義済みの色、サイズ、形などの組合せを示すテストページを出力します。コマンド `set linetype` を使うことでこれを変更したり、あらたな組を追加定義できます。

### 3.6 グラフをアニメーション化できますか

**gnuplot** 出力形式で直接アニメーションファイルを出力できるのは以下のものだけです:

```
set terminal gif animate {delay <time>} {loop <N>} {optimize}
```

デモの中の <http://gnuplot.sourceforge.net/demo/animate.html> も参照してください。

### 3.7 どうやったら陰関数のグラフが書けますか

**gnuplot** では陰関数のグラフや曲線は直接は描画できません。しかし、次善の策はあります。

```
gnuplot> # 例: 次の行を適当な定義に置き換えてください。
gnuplot> f(x,y) = y - x**2 / tan(y)
gnuplot> set contour base
gnuplot> set cntrparam levels discrete 0.0
gnuplot> unset surface
gnuplot> set table 'curve.dat'
gnuplot> splot f(x,y)
gnuplot> unset table
gnuplot> plot 'curve.dat' w l
```

その仕掛けは、曲面  $z=f(x,y)$  の 1 本の等高線  $z=0$  を描き、その等高線のデータをファイルにセーブすることにあります。

### 3.8 2つの曲線間の領域を塗りつぶすには

2つの与えられた曲線の間を塗りつぶす描画は、疑似ファイル '+' と *filledcurves closed* を使えば簡単にできます。以下の例は、2つの曲線  $f(x)$  と  $g(x)$  に対するデモです:

```
f(x)=cos(x)
g(x)=sin(x)
xmax=pi/4
set xrange [0:xmax]
plot '+' using 1:(f($1):(g($1)) with filledcurves closed
```

注意してもらいたいのですが、上の例は2つの曲線間の領域を塗りつぶすものであって、不等式  $g(x) < f(x)$  を満たす領域を塗りつぶすものではありません。もし、後者を行ないたいなら、*using* 部分で3項演算子を使って、その不等式が満たされない場合は未定義値 (0/0) を返すようにする必要があります。

*textit*help *filledcurves*, *help special-filenames*, *help ternary* に関するドキュメントを参照してください。*demos* ディレクトリ内の *fillbetween.dem* も参照してください。

### 3.9 pm3d によるデータファイルの splot で何も表示されません

`set pm3d; splot 'a.dat'` とやっても、カラーボックスは表示されるもののグラフは表示されないとしたら、もしかしたら2つのつながっている線(孤立線)を分離する空行がないのでは? 空行を入れなさい! これが何を意味するのかに興味があるなら、gnuplotのdemoディレクトリにある *demo/glass.dat* や *demo/triangle.dat* などのファイルをじっくり見てください。

以下のawkスクリプト(例えば *addblanks.awk* と呼ぶことにします)は有用で、データファイルに、第1列目の数を変更したときに空行を追加してくれます。

```
/^[[:blank:]]*#/ {next} # コメント行 (# で始まる行) は無視
NF < 3 {next} # 3 列未満しかない行も無視
$1 != prev {printf "\n"; prev=$1} # 空行を出力
{print} # その行自体を出力
```

これにより、データファイルをコマンド `awk -f addblanks.awk <a.dat` によって前処理するか、unix互換のプラットフォーム上でデータファイルを

`gnuplot> splot "<awk -f addblanks.awk a.dat"` のように描画するか、のどちらかを行ってください。

### 3.10 3次元データの2次元射影を描画するには

`set view map` を使用してください。2次元のカラー画像の描画用には、描画スタイル `with image` や `with rgbimage` なども用意されています。

### 3.11 pm3d 色地図/曲面上に dots/points の散布図を重ねるには

pm3dスタイルの `explicit` スイッチ (`implicit` も参照のこと) を使ってください:

```
gnuplot> set pm3d explicit
gnuplot> splot x with pm3d, x*y with points
```

### 3.12 ラベル付きの等高線を描画するには

個々の等高線にラベルをつけたグラフは、gnuplot バージョン 4 ではある種の裏技と余分な処理手順が必要です。以下を参照してください: [gnuplot.sourceforge.net/scripts/index.html#tricks-here](http://gnuplot.sourceforge.net/scripts/index.html#tricks-here)

バージョン 5 では、その手順はとても簡単です。等高線を、1 回目は "with lines" で、2 回目は "with labels" で 2 回描画するだけです。目立つラベルを作成したい場合は、以下のような設定を使用するといいでしょう。

```
set style textbox opaque noborder
set contours
splot 'DATA' with line, 'DATA' with labels boxed
```

### 3.13 pm3d 色地図/曲面グラフに等高線を重ねるには

[既に不要?] これは、等高線を table 出力形式を利用して一時ファイルに書き出し、そしてそれを set contours を使わずに描画する必要があります。次の例は色地図グラフのデモですが、曲面グラフの場合は set pm3d map を削除して set ticslevel 0 を入れます。

```
# 関数 x*x-y*y の等高線を一時ファイルに書き出す
set contour base; set cntrparam level 20
unset surface
set table 'contour.dat'
splot x*x-y*y
unset table

# 1 行の空行を 2 行の空行に変換
!awk "NF<2{printf\"\n\n\"}{print}" <contour.dat >contour1.dat

# グラフの描画
reset
set palette gray
set palette gamma 2.5
set pm3d map
set pm3d explicit
splot x*x+y*y with pm3d, 'contour1.dat' with line lt -1
!rm contour.dat contour1.dat
```

最後のコマンドは 2 つの一時ファイルを削除します。

### 3.14 pm3d による色付きの多面体描画は

3 次元物体の面に色をつけて描画するには、ファイルを以下のような形式で作成すれば可能です。

```
# 三角形 1
x0 y0 z0 <c0>
x1 y1 z1 <c1>

x2 y2 z2 <c2>
x2 y2 z2 <c2>

# 三角形 2
x y z
...
```

1 行の空行と 2 行の空行の位置に注意してください。<c> はオプションの色指定です。これを以下のようにして描画します (どちらかの splot で):

```
set pm3d
set style data pm3d
splot 'facets.dat'
splot 'facets_with_color.dat' using 1:2:3:4
```

曲面の線は、`set style data pm3d` か、`splot ... with pm3d` で無効にできることに注意してください。

上の例では、`pm3d` は三角形をそれぞれ独立な曲面として描画します。それらは、データファイル中に他のものが見つかったときにその後で一つの曲面に描かれます。重なる部分は2次元射影では重なって描画されてしまいます

`gnuplot` は3次元モデリングプログラムではなく、隠線処理ルーチンは点、線には適用されますが、面には適用されません。面の集まりとしてのデータの処理なしには、影に隠れる面を作ることはできません。'hidden3d' アルゴリズムは、入力データを2つの方法で使って実行されています: 一つは、曲面を構成する三角形(格子点の四角形から作られる)の集まりをセットアップすること、もう一つは端の集まりです。そしてそのすべての端に対して、どの部分が他の面の後ろに隠れないのかをチェックし、そしてそれらを描画しています。

結果として、`gnuplot` は曲面や3次元の物体を仮想現実としては描画しません。これは `set pm3d map` に対してはちゃんと働きますが、本当の3次元にしたいなら、多面体データをVRMLファイルに変換するソフトを書く方がいいかもしれません。

### 3.15 色地図をモノクロのプリンタでもちゃんと印字するようなパレットは

`set palette cubehelix` を試してください。

## 4 欲しい機能 (Wanted features)

### 4.1 `gnuplot 4.6, 5.0` 等で新しくなったことは

ここにあげるには多すぎます。ソース配布物に含まれるファイル `NEWS`、あるいは `gnuplot` のヘルプドキュメントの "New features" のセクションを参照してください。

### 4.2 `gnuplot` は画像フォーマット用のドライバをサポートしていますか

インストール済の `gnuplot` が使用できるグラフィックドライバの一覧を見るには、`set term` とタイプしてください。

通常の配布物には含まれていても、デフォルトでは使えないようになっているグラフィックドライバもあります。それらが使いたければ、`gnuplot/src/term.h` を修正してコンパイルし直す必要があります。

### 4.3 `gnuplot` は隠線処理できますか

ええ。

### 4.4 `gnuplot` は棒グラフ/ヒストグラム/箱型グラフが書けますか

`gnuplot` は、あらかじめ表形式にされたデータを表示するための、さまざまな横並び型 (clustered) と積み上げ型 (stacked) のヒストグラムスタイルをサポートしています。また、生のデータを場所毎に累積するためのいくつかのオプションも提供していて、それを切り替えて棒グラフとして表示させることも可能です。マニュアルの *smooth frequency* の部分を参照してください。

#### 4.5 gnuplot は円グラフが書けますか、四半期グラフはどうですか

円グラフは `gnuplot` ではちょっと面倒ですが、以下を見てください。 <http://gnuplot.sourceforge.net/demo/circles.html> または、以下を参照してください。 <http://gnuplot-tricks.blogspot.com/2009/08/pie-charts-entirely-in-gnuplot.html>

デモの中には単純なガントチャートの例があります。これは、`gnuplot` バージョン 5 の新しい機能です。

#### 4.6 一つのページに複数のグラフが書けますか

はい。 `set multiplot` としてください。

#### 4.7 gnuplot は一つのグラフで複数の y 軸を使えますか

はい。2次元グラフでは別々に x 軸を下 ( $x_1$ ) と上 ( $x_2$ ) に持つことができますし、別々に y 軸を左 ( $y_1$ ) と右 ( $y_2$ ) に持つことができます。バージョン 5 では、*with parallelaxes* という描画モードで y 軸をさらに追加することもできます。

#### 4.8 一つのファイルに gnuplot のコマンドとデータの両方を書けますか

バージョン 5 では、「ヒアドキュメント」形式での名前付きデータブロックをサポートしています:

```
gnuplot> $DATABLOCK << EOD
cats 4 2
dogs 1 4
EOD
gnuplot> plot $DATABLOCK using 2:3:1 with labels
```

名前付きデータブロックは一度定義すれば、それを何度でも使うことができます。

データは、`plot` コマンド上で疑似ファイル指定 `plot "-"` を行ってインライン形式で与えることもできます。この場合は、データは一度しか使えません。

```
gnuplot> plot "-"
1 1
2 4
3 9
e
```

#### 4.9 ラベルにギリシャ文字や上/下付きの文字を使えますか

多くの出力形式 (出力デバイスドライバ) で、「拡張文字処理」モード ("enhanced text" モード) がサポートされていて、これで上付き下付きの文字が扱えます。その場合、あなたのシステムにインストールされているフォントのサポートする範囲内で、ギリシャ文字や、数学記号文字も扱えます。色んな特殊文字を入力する最も簡単な方法は、あなたのシステムがサポートしていればですが、UTF-8 エンコーディングを利用することです。これにより、フォントの変更の必要がなくなります。

いくつかの  $\text{\LaTeX}$  系出力形式 (*latex*, *epslatex*, *tikz*, *context*, *cairolatex*) は、生成した文字列を  $\text{\LaTeX}$  に渡します。この場合は、`"\alpha_{3}"` や `'\alpha_{3}'` のような通常の  $\text{\LaTeX}$  風の記述が利用できます。

#### 4.10 どうしたらアクセント付きの文字を出力できますか

ü (u ウムラウト; u の上に点 2 つのアクセント) や ñ のような文字をラベルに含めるには、適切なエンコードオプションを設定し、8bit 文字コードを使用しなければいけません。例えば次のようにします。

```
gnuplot> set encoding iso_8859_1
gnuplot> set title "M\374nchner Bierverbrauch \374ber die Jahre"
gnuplot> plot "bier.dat" u 1:2
```

これにより、適切な *set encoding* により、チェコ語、仏語、ハンガリー語、露語などのラベルを打ち出すことができます。しかし、2 種類のエンコード (例えば西ラテンエンコードと東ラテンエンコード) を一つのファイルに混ぜることはできません。

UTF-8 エンコーディングフォントを使うためのより一般的なやり方は、**gnuplot** 内で直接 UTF-8 の文字を入力することです。これは多くの出力形式で動作しますが、しかし PostScript ではとても厄介です。

#### 4.11 縦と横の軸を 1:1 にできますか

`set size square` か `set view equal xy` としてみてください。

#### 4.12 異なるサイズのテキストを一つのグラフで出力できますか

たいていの出力形式では、デフォルトフォントの名前とサイズを指定できます。「拡張文字処理」モード ("enhanced text") では、フォント、および文字サイズやボールド、イタリックの書体を一つのグラフ内で変更できます。

#### 4.13 どうしたらデータ点をスキップできますか

以下のようにデータ値として ? を指定すればできます。

```
1 2
2 3
3 ?
4 5
```

*set missing* も参照してください。データファイルのコメント文字を指定するには、*set datafile commentschars* を参照してください。

#### 4.14 どうしたら n 点毎にデータを扱えますか

これは、例えば `plot 'a.dat' every 2` のように、`plot` コマンドに色々なオプションを指定することでできます。すべての点を通る線を引き、しかし点の記号は n 点毎に書きたい場合は、`plot 'a.dat' with linespoints pointinterval n` としてみてください。

#### 4.15 どうしたら鉛直線を書けますか

状況によりますが、主な方法は以下の通りです:

- `set arrow .... nohead` この場合、矢の正確な開始点と終了点の位置を計算しておく必要があります。
- データ点を作って (インラインでの指定も可) それを描画

## 4.16 どうしたらデータファイルを描画できますか

簡単です。コマンド `plot 'a.dat'` でできます。3次元描画の場合は `splot 'a.dat'` です。ただし、2つのつながっている線(孤立線)を分離するのに空行を出力するのを忘れないでください。さもないと、データが格子状(grid)でないというエラーが出るかも知れません。質問 3.9 も参照してください。データが格子状でない場合は、`set dgrid3d {色んなオプション}` を使用してください。

## 4.17 どうしたら `multiplot` の描画を `replot` できますか

直接はできません: `gnuplot` は `replot` コマンドはサポートしていますが、`remultiplot` コマンドはありません。よって、スクリプトファイル内に `set multiplot` から `unset multiplot` までのすべてのコマンド列を書く必要があります。そのスクリプトを `gnuplot` に `load` してやれば、必要な回数だけ、異なる出力形式、異なる出力ファイルに `replot` できます。

## 5 その他

### 5.1 バグを見つけたのですがどうしたら良いですか

最初に、それが本当にバグなのか、それともある `set-command` で翻すことができる特徴なのかを見極めてください。

次に、あなたが使っている `gnuplot` が古い版ではないか確かめてください。もしそうなら、より新しい版ではその新しいバグは取り除かれている可能性があります。

CVS 開発版は、現在のリリース版以後に報告されたバグの修正を既に持っている可能性があります。レポートを送る前に、質問のバグが既に修正されていないかチェックしてみてください。

これらをチェックした後でなおそれがバグであると確信したなら、以下に進んでください。もしあなたが公正で普遍的な種類のバグレポートを作ったならば、多分ニュースグループ [comp.graphics.apps.gnuplot](http://comp.graphics.apps.gnuplot) への投稿が取るべき道でしょう。もし問題の詳細な調査結果があるならば、特にそのエラーを際限する簡単なスクリプトを提供できるならば、それを [sourceforge.net/tracker/?group\\_id=2055&atid=102055](http://sourceforge.net/tracker/?group_id=2055&atid=102055) の bug-tracker にアップロードしてください。

sourceforge の tracker は、バグのレポート、および後のリリースに反映するためのバグの修正の収集のためのものです。ニュースグループ [comp.graphics.apps.gnuplot](http://comp.graphics.apps.gnuplot) は `gnuplot` に関する問題に対する回避策を見つける、あるいはそれを実際に解決する手助けのためのものです。もしバグレポートを送るときは、`show version long` コマンドで見れる `gnuplot` のバージョン(パッチレベルも含めて)、出力形式のドライバ名 (terminal driver)、オペレーティングシステム、バグの正確な記述とバグを再現するための入力、これらを確認しレポートに入れてください。これらの詳細が示されていないければ、その問題の解を与えることはほとんど不可能です。そして、最新の公式版の `gnuplot` に対して当てた差分 (context diff) も、可能ならばすべて引用すべきです。

### 5.2 私のプログラムで `gnuplot` のルーチンを利用できますか

パイプをサポートしているシステムでは、他のプログラムからパイプ経由で `gnuplot` にコマンドを渡すことができます。Octave ([www.octave.org](http://www.octave.org)) のように、`gnuplot` をグラフィックエンジンとして使用している多くのアプリケーションがこの方法を使用しています。これは、form ベースの Web ページから `gnuplot` を起動する cgi スクリプトでも利用できます。

John Campbell ([jdc@navvax.ucc.nau.edu](mailto:jdc@navvax.ucc.nau.edu)) は、かなり前の版 (3.5) の `gnuplot` を改良して、C プログラムから呼び出せる C のサブルーチンライブラリに作り上げました。`gnuplot` はその後劇的に変化しましたが、現在の版に基づく同様のライブラリを作成するような計画は我々は知りません。

### 5.3 どう追加や拡張がなされていますか、それはどう手に入れますか

拡張やパッチは、gnuplotの開発サイト [sourceforge.net/tracker/?group\\_id=2055&atid=302055](http://sourceforge.net/tracker/?group_id=2055&atid=302055) の "Patches" ページにあります。一般的に、現在の開発版は、最終公式リリース版の gnuplot にはまだ入っていない機能も入っています。

### 5.4 積分や FFT や IIR フィルタなどが欲しいのですが

gnuplot は今までも、そして現在も描画プログラムであり、データ処理や数式処理プログラムではありません。よってそのようなことはできません。積分の基本的な実装については demo の "bivariat.dem" を見てください。しかし、gnuplot バージョン 5 は、動的外部共有オブジェクトの関数の呼び出し、いわゆる「プラグイン」をサポートしています。よって、C あるいは他の言語による複雑な関数を、決められた呼び出し様式に従って書けば、gnuplot がインポートできるプラグインとなるようコンパイルすることができます。

### 5.5 "heavy-duty" データを gnuplot で使えますか、または gnuplot に変わるものはないですか

gnuplot それ自身は、しんどい数値計算にはあまり向いていません。しかし、かなり大きなデータ集合の処理は可能です。

Martin-D. Lacasse が書いた、対話型の多目的曲線回帰プログラム *fudgit* を試してみてください。これは、グラフ出力用に内部で gnuplot を使います。

NCSA (National Center for Supercomputing Applications) の STG (Software Tools Group) によって開発されたアプリケーションもあります: [ncsa.uiuc.edu](http://ncsa.uiuc.edu)。

pgperl を試してもいいでしょう。これは Perl 5 によって PGPLOT プロットパッケージを統合したものです。これに関する情報は [www.ast.cam.ac.uk/AAO/local/www/kgb/pgperl](http://www.ast.cam.ac.uk/AAO/local/www/kgb/pgperl) で見つかりますし、ソースは [ftp.ast.cam.ac.uk](http://ftp.ast.cam.ac.uk/pub/kgb/pgperl/) の /pub/kgb/pgperl/、または [linux.nrao.edu](http://linux.nrao.edu/pub/packages/pgperl/) の /pub/packages/pgperl/ にあります。

他の方法として Octave があります。README を引用します: Octave は高水準言語で、主に数値計算用です。Octave は GPL ライセンスに従い、原則的にフリーな Matlab クローンです。線形や非線形の問題を数値的に解くのに便利なコマンドラインインターフェースを備えています。Octave の最新版は常に [www.octave.org](http://www.octave.org) にあります。ところで、octave は gnuplot を描画エンジンとして使いますので、gnuplot に追加するデータ処理プログラムを手に入れることになります。

最後に、*matlab* の代わりとして、オープンソースの *scilab* ([www.scilab.org](http://www.scilab.org)) もあります。

### 5.6 私の対話型ターミナルでどのようにホットキーを使えばいいですか

ほとんどの対話型出力形式では、再描画、各要素の描画のトグル、軸のスケールの変更などを行うために、あらかじめ設定された、およびユーザ定義可能なホットキーの両方をサポートしています。現在有効な gnuplot の描画ウインドウ上で *h* を打つと、ホットキーの一覧が表示されます。さらなる情報については、*help mouse* と *help bind* を参照してください。

### 5.7 gnuplot を他のシステムに移植、またはパッチを作りました。どうしたらいいですか

投稿よりもむしろ、[sourceforge.net/tracker/?group\\_id=2055&atid=302055](http://sourceforge.net/tracker/?group_id=2055&atid=302055) の 'Patches' セクションを通してコメントするかパッチを更新する方が望ましいです。[gnuplot-beta@lists.sourceforge.net](mailto:gnuplot-beta@lists.sourceforge.net) は、より活発な議論のための論点を送って頂けるとありがたいです。

## 5.8 gnuplot の次期バージョンの開発について知りたいのですがどうしたらいいですか

gnuplot のベータテストメーリングリストに入ってください。メール本文(メールタイトルではなく)に `subscribe gnuplot-beta` と書いてメールを [Majordomo@lists.sourceforge.net](mailto:Majordomo@lists.sourceforge.net) に送ってください。

## 5.9 FAQ に含まれるような未解決の質問は？

その質問を(回答と一緒に) [gnuplot-beta@lists.sourceforge.net](mailto:gnuplot-beta@lists.sourceforge.net) へ送ってください。

# 6 うまく扱うには (Making life easier)

## 6.1 どうしたら 2 つの関数を重ならないように書けますか

これは、以前は複雑でしたが、バージョン 5 ではこれは簡単です。描画する各関数の直前に必要な範囲を置きます。例えば、実験データと、領域の 2 つの異なる部分をカバーする異なる 2 つの異なる関数モデル  $f_1, f_2$  を描画するには:

```
gnuplot> set autoscale x # x の範囲をデータから取得
gnuplot> plot "data", [-100:0] f1(x), [0:100] f2(x)
```

## 6.2 どうしたら描画前にデータにフィルタをかけられますか

そのシステムが、Unix のように `popen()` 関数をサポートしているなら、他のプロセスの出力を使って実行することが可能です。例えば、次のような `awk` の短いプログラムが使えます:

```
gnuplot> plot "< awk ' { print $1, $3/$2 } ' file.in"
```

`plot` コマンドはとても強力で、データファイルに対するある種の代数処理も可能です。`help plot` を参照してください。

上のフィルタは、Unix やその互換 OS、OS/2 では問題なく動作します。Windows では、これはデフォルトでは `gnuplot` バージョン 5 のみがサポートしますが、その前のバージョンでは、別のプログラム `wgnuplot_pipes` が必要になります。

## 6.3 どうしたら現在の設定を保存/再利用できますか

それには `save` と `load` のコマンドを使ってください。詳細は `help save` と `help load` を参照してください。

`set term push` と `set term pop` を使えば、ファイルシステムを触ることなくそれぞれ現在の出力形式の保存とその後の復帰ができます。

## 6.4 どうしたら `splot` で曲線(格子でない)を書けますか

`splot` への入力データが、各行が同じデータ点数(通常は空行で分離されている)を持つように揃っていれば、`splot` はデフォルトではそのデータを曲面を記述するものとして扱います。そうでなく個々の曲線を描画したければ、`unset surface, set surface explicit, plot ... nosurface`などを組み合わせてみてください。

## 6.5 どうしたら x-y 平面上で他の関数に制限されている $f(x,y)$ のグラフを書けますか

以下は一つの方法です:

```
gnuplot> f(x,y) = x**2 + y **2
gnuplot> x(u) = 3*u
gnuplot> yu(x) = x**2
gnuplot> yl(x) = -x**2
gnuplot> set parametric
gnuplot> set cont
gnuplot> splot [0:1] [0:1] u,yl(x(u))+(yu(x(u)) - yl(x(u)))*v,\
> f(x(u), (yu(x(u)) - yl(x(u)))*v)
```

## 6.6 どうしたらプログラムから gnuplot を呼び出せますか

unix 互換システム上ならば、gnuplot へのコマンドを標準入力 (stdin) からパイプで渡せます。gnuplot からの print コマンドは、名前つきパイプで読み出すことができます。Windows では、GUI プログラムで標準入力 (stdin) が使えないため、コンソール版の *gnuplot* (推奨) か、テキストコンソールが追加された GUI 版の *wgnuplot\_pipes* のいずれかを使う必要があります。古い *pgnuplot* というヘルパープログラムもまだ配付パッケージに含まれています。

## 6.7 Planck 定数 ( $h$ に横線) が必要なときはどうしたらいいですか

最も自然な方法は、UTF-8 フォントを使い、文字  $h$  (Unicode コードポイント #x210F) を直接入力することです。

しかし、これは PostScript ではうまくいきませんので、例えば次のような次善の策を用いる必要があります: `@{/=56 -} {/=24 h}` または `{/=8 @{/Symbol=24 -} _{/=14 h}` 後者では 24pt (ポイント) の "-" (/Symbol フォントの中の長いハイフン) がスペースなしで使われます。14pt の "h" は 8pt のスペース ("\_" の前にあるスペース) だけずらされて置かれますが、それは下付き文字として書かれるので小さい文字になります。しかしそれらは通常我々が使用するプランク定数の記号とは違って、横線は斜線ではなく水平線になります。アクセント記号なども使ってみました (iso-latin-1 エンコーディングの文字 264 など)、その大きさと位置を正しくする方法を見つけることはできませんでした。もう一つの方法は `{/=14 @^{/Symbol=10 -}{/=14 h}` とすることです。

AMS-LaTeX の PostScript フォント ([www.ams.org/tex/amssymbols.html](http://www.ams.org/tex/amssymbols.html)) にあり、そして多くの LaTeX 配布にも含まれています) を使えば、とても簡単に小さい Planck 定数をセットすることができますが、gnuplot (help fontpath を参照) と、PostScript インタープリタ (通常 Ghostscript) がファイル `msbm10.pfb` (または `msbm10.pfa`) がどこにあるかを知る必要があります。"h" に傾斜のついた横棒を重ねた `\hslash` を生成するには `{/MSBM10 \175}` を使ってください。通常の `\hbar` (水平の横棒) は 8 進数の 176 です。ただし、水平の横棒はイタリック体としてしか存在しないことに注意してください。

## 6.8 太陽のような数学記号 (丸に点) が必要なときはどうしたらいいですか

Planck 定数と同様、最も自然な方法は UTF-8 フォントを使い、文字  $\odot$  (Unicode コードポイント #2299; "circled dot operator") を直接入力することです。非常によく似たグリフがコードポイント #x2609 にもあります。これは "sun" で、よりいいかもしれませんが、それを持つフォントは多くはありません。

## 6.9 全く何もないページを出力するには

うーん、多分それは全く何も出力したくないわけではなくて、タイトルだけのものを出力したいじゃないでしょうか (multiplot のページの他のグラフに重ねるような):

```
reset; unset xtics; unset ytics
unset border; unset key
set title 'Title on an empty page'
plot [][0:1] 2
```

## 6.10 境界の位置を丁度ある場所に合わせてグラフを作るには

上、下、左、右の境界位置を、ページ内の位置として小数単位で指定してください:

```
set lmargin at screen 0.05
set bmargin at screen 0.05
set rmargin at screen 0.95
set tmargin at screen 0.95
```

## 7 共通の問題

### 7.1 助けて! 私のフォントが使えない。

gnuplot は、それ自身ではフォント処理を行いませんので、それは必然的に個々のデバイスをサポートするライブラリに任せることになります。残念ながらこれは、フォントの検出においては、異なる出力形式には異なる説明が必要だということを意味しますので、ここでは、簡単なヒントをいくつか上げておきます。詳しい情報については、あなたが問題にしている出力形式用の gnuplot のドキュメントを参照してください。

**png/jpeg/gif** これらの出力形式は、libgd というサポートライブラリを使用し、これは環境変数 GDFONT-PATH で与えられるディレクトリ内にフォントを探します。そのように libgd のフォントパスを設定すると、今度は gnuplot のデフォルトフォントを設定したいかもしれません。例:

```
setenv GNUPLOT_DEFAULT_GDFONT verdana
```

**pdf** このサポートライブラリ libpdf は、関連するフォント設定ファイルを参照します。通常それは /usr/local/share/pdflib.upr としてインストールされます。環境変数 PDFLIBRESOURCE は、このファイルの位置を示します。

**post** PostScript フォント名は、その文書が印刷されるまで参照されませんから、あなたのプリンタでどのフォントが有効であるかを gnuplot は知りません。よって、どんなフォント名でも指定することができます。しかし、gnuplot の出力にフォントを埋め込むことは可能です。それについては、gnuplot の内部コマンド “help set term post fontfile” で得られる指示を参照してください。

**svg** フォント処理は、ビューワ依存です。

**x11** x11 出力形式は、通常の x11 フォントサーバ機構を使用しますが、少しだけトリッキーな点があり、マルチバイトフォントを使用するためには、明示的に以下のように指定する必要があります:

```
set term x11 font "mbfont:sazanami mincho,vera,20"
```

**win** ツールバーの "Options" のプルダウンメニューから "Choose font..." を選択します。

**wxt, qt** Linux システムでは、これらの出力形式はシステムの fontconfig ユーティリティが提供するフォントを使用します。

## 7.2 x11 または qt で gnuplot が描画ウィンドウを開きません。なぜですか

VMS 上ではいくつかの記号を作らなければいけません:

```
$ gnuplot_x11 := $disk:[directory]gnuplot_x11
$ gnuplot := $disk:[directory]gnuplot.exe
$ def/job GNUPLOT$HELP disk:[directory]gnuplot.hlb
```

そして **gnuplot** をコマンドラインから起動し、`set term x11` としてください。

Unix 上では、x11 と qt 出力形式は、分離されたヘルパープログラム *gnuplot\_x11*, *gnuplot\_qt* のインストールが必要です。これらは通常は */usr/libexec/gnuplot/5.0/* というディレクトリにインストールされ、**gnuplot** はその場所を探すことを知っています。しかし、それ以外の場所にインストールしたり、それらを見つける際に問題が起きる場合は、環境変数 `GNUPLOT_DRIVER_DIR` を設定してみてください。

## 7.3 なぜ gnuplot はとても小さい数字のデータを無視するのですか

**gnuplot** は、ある目的のため  $1e-08$  より小さい数を 0 として扱います。よってとても小さい数の集まりを描画しようとするときそれらは 0 として描画されてしまいます。対数軸で描画する場合は、もっとひどいことにそれらは目盛りからはずれてしまいます。また、全ての数が "0" である場合、範囲は空であるとみなされます (訳注: **gnuplot** 3.6 以前の仕様):

```
gnuplot> plot 'test1'
Warning: empty y range [4.047e-19:3e-11], adjusting to [-1:1]
gnuplot> set yrange [4e-19:3e-11]
gnuplot> plot 'test1'
      ^
y range is less than `zero`
```

その対策は **gnuplot** の "zero" の概念を変更することです。

```
gnuplot> set zero 1e-20
```

より詳しい情報については `help set zero` を参照してください。

## 7.4 シェルやスクリプトから gnuplot を実行しても結果のグラフはスクリーン上で一瞬出るだけでそのあと出てきません

1. ファイル内の `plot` コマンドの後、またはファイルの最後に `pause -1` を入れる方法があります。
2. スクリプトの終了時に対話型部分にいるようにするには `gnuplot filename.gp -` (最後のパラメータはマイナス) を使用します。
3. **gnuplot** を `gnuplot -persist` と実行する方法があります。
4. Windows 上なら、`-persist` か `/noend` のどちらでも結構です。
5. コマンド `set terminal` の一部分としてオプション `persist` を与えても結構です。

## 7.5 書いた式 (1/3 など) が意味のない結果を出力してしまいます

**gnuplot** は整数の式の場合、実数演算ではなく整数演算を行ないます。例えば式  $1/3$  は 0 と評価されます。もし実数の値が欲しいならば、その数の後にドット "." をつけてください。例:

```
gnuplot> print 1/3
      0
gnuplot> print 1./3.
      0.333333
```

整数の式をこのようにして評価するやり方は C や Fortran でも行なわれています。

## 7.6 私の出力が不完全です!

`set output` によって出力をクローズして、たまっている出力を吐き出させる必要があるかも知れません。出力形式によっては、一つの出力ファイルに複数のページが含まれています (`postscript`, `pdf`, `latex`, `svg`, ...)。これらの出力モードでは、`gnuplot` はあなたが描画の追加ができるように各描画後にファイルをオープンしたままにしています。よってあなたが明示的にそれをクローズする (`set output` または `unset output`) か、異なる出力形式を選択する (`set term`) か、または `gnuplot` を終了するまでは、そのファイルは完全ではなく、外部のアプリケーションでは使用できるようにはなっていません。単一 'ページ' のみを持つ出力形式 (`png`, `emf`, ...) ではこのような問題に悩まされることはないでしょう。

## 7.7 $\LaTeX$ -terminal の出力は $\LaTeX$ の実行でエラーが出ます

裸の  $\LaTeX$  2<sub>ε</sub> は今はコマンド `"\Diamond"` も `"\Box"` も持っておらず、それらは今は `latexsym` パッケージに入っています。また、`amssymb` パッケージの他の記号も使います。これらはいずれも  $\LaTeX$  の基本配布物には含まれていて、よって任意の  $\LaTeX$  システムには含まれています。これらのパッケージをあなたの文書に入れることを忘れないください。

## 7.8 パイプで `gnuplot` を呼び出したり `gnuplot` スクリプトを使ってグラフが書けません

次のような短い Perl スクリプトを使って `gnuplot` を呼び出せます:

```
#!/usr/local/bin/perl -w
open (GP, "|/usr/local/bin/gnuplot -persist") or die "no gnuplot";
# force buffer to flush after each write
use FileHandle;
GP->autoflush(1);
print GP, "set term x11;plot '/tmp/data.dat' with lines\n";
close GP
```

`gnuplot` は終了時にその描画ウィンドウを閉じます。 `close GP` コマンドが実行されると、描画ウィンドウを目にする前でさえも閉じられてしまいます。

それに対する対処は3つあります: 1つ目は、`gnuplot` 内で `pause -1` コマンドを、パイプを閉じる前に使用すること、2つ目は、`gnuplot` と描画ウィンドウが不要になったときのみパイプを閉じるようにすること、3つ目はコマンドラインオプション `-persist` を使うことです。このオプションは X-Window System に描画ウィンドウを開いたまま残します。

## 8 帰属

`gnuplot` 3.7 の主な寄与者は (アルファベット順で)、Hans-Bernhard Broecker, John Campbell, Robert Cunningham, David Denholm, Gershon Elber, Roger Fearick, Carsten Grammes, Lucas Hart, Lars Hecking, Thomas Koenig, David Kotz, Ed Kubaitis, Russell Lang, Alexander Lehmann, Alexander Mai, Carsten Steger, Tom Tkacik, Jos Van der Woude, James R. Van Zandt, Alex Woo です。さらに version 4.0 への重要な寄与者として、include Ethan Merritt, Petr Mikulík, Johannes Zellner がいます。version 4.2, 4.4, 4.6, 5.0 のリリースは Ethan Merritt が取りまとめました。

このリストは Russell Lang, John Campbell, David Kotz, Rob Cunningham, Daniel Lewart, Alex Woo の寄与を John Fletcher が最初にまとめたものです。そして、Alex Woo の草案により、Alex Woo, John Campbell, Russell Lang, David Kotz らの修正と追加、Daniel Lewart による多くの修正が Thomas Koenig によりまとめ直されました。そして、`gnuplot` 3.7 用に再び Lars Hecking, Hans-Bernhard Broecker およびその他の人々による修正が Alexander Mai と Juergen v.Hagen によりまとめ直されました。`gnuplot` 4.0 用の修正は、Petr Mikulík と Ethan Merritt により行なわれました。`gnuplot` 4.2 用の修正は、Petr Mikulík と Ethan

Merritt により行なわれました。**gnuplot** 4.4 と 4.6 用の修正は、Ethan Merritt により行なわれました。  
**gnuplot** 5.0 用の修正は、Ethan Merritt により行なわれました。  
(訳注: 日本語訳は Shigeharu TAKENO ([shige@iee.niit.ac.jp](mailto:shige@iee.niit.ac.jp)) により行なわれました。)